

Collimator transmission

Project work in the course Modern Experimental Physics

This project work is a part of the course, experimental modern physics. Course coordinator: Joseph Nordgen. It has been undertaken at the department of radiation sciences at Uppsala University. It is a four-week assignment where the student is supposed to get a view of modern physics. Project involves editing of a computer program in Fortran90 and optimizing some of the parameters. It also involves the running of computer calculations and analyzing the results.

Abstract

When doing tomography on spent nuclear fuel the activity distribution is reconstructed via measurements and calculations. Emitted radiation intensities are measured at a great number of points around the fuel, these are stored as a vector I . Through the following vector relation the activity distribution A can be determined if W is known.

$$\overline{W} \cdot \overline{A} = \overline{I}$$

W is a matrix depending on the geometry of the fuel and detector set up.

To shield the detector from radiation a collimator of a tungsten alloy is used. Radiation from the fuel is supposed to be stopped by the collimator except in a slit leading to the detector. In reality a small part of the radiation is transmitted through the edges of the collimator into the slit and reaches the detector. This contribution is represented by an integral in the matrix elements mentioned above. A program for calculating this integral already existed, but it had to be modified and optimized.

By investigating how the results of the calculations of the collimator transmission varied with different step length in the integration over

the detector and the fuel, an optimized set of step lengths was achieved.

The contribution from the collimator transmission was calculated for the following gamma energies, 662 keV, 757 keV, 1596 keV and 2186 keV.

One question to be answered was whether a point on the horizontal plane (z -plane) is representative for the whole transmission from a vertical rod through that point. It was investigated using a quotient between the value at $z=0$ and the total transmission. Results showed that it is not the case.

The collimator transmission showed different behavior in three distinct regions, slit, penumbra and umbra region. A fit of an analytical expression was made to each region. In the slit and penumbra regions the fit was satisfying, but in the umbra region the fit discrepancy was relatively large. This is because the program used to make the fits does not have the complex functions needed to make a fit in this region.

Marcus Agåker
Semanders v3:632
752 61 Uppsala

tel: 018-461694
073-6312324

Table of contents

Abstract	1
1 Introduction	3
2 Theory	4
3 Simulations	6
3.1 Modification of the original program	6
3.2 Reference simulation	6
3.3 Optimization of numerical integration steps.....	7
3.4 Optimized simulation and fitting of analytical expressions to 3D surfaces.....	7
3.6 Investigations with different gamma ray energies	7
4 Results	8
4.1 Collimator transmission in reference simulation	8
4.2 Quotient of transmission at $z_{\text{fuel}}=0$ to total.....	9
4.3 Optimization of numerical integration over detector surface	9
4.4 Optimization of numerical integration over fuel	10
4.5. Selected set of numerical integration steps.....	11
4.6 Collimator transmission in optimized simulation.....	12
4.7 3D-surfaces.....	13
4.9 Collimator transmission for various gamma-ray energies	15
5 Conclusion and Discussion.....	17
6 Appendix A	18

1 Introduction

When spent fuel elements in a nuclear power plant are replaced it is of interest to experimentally verify various fuel parameters e.g. power distribution and integrity of the fuel. One method to do this is under development at the department of radiation sciences. By using tomography, the fuel rods can be inspected easily without having to disassemble the fuel, which is expensive and dangerous. Tomography is a method normally used in medicine to give a two dimensional picture of the cross section of a patient. Emitted radiation intensities are measured at a large number of locations around the object. These intensities can be used to reconstruct the complete activity distribution.

The nuclear fuel is highly radioactive. To protect the sensitive detectors from the exposure of too much radiation, a collimator is used, i.e. a big solid block of a dense material in which a small slit has been made see fig 1.1.




Figure 1.1 Collimator block

The detector is mounted behind the slit and the only radiation to reach the detector is that which passes through the slit. In order to stop as much unwanted radiation as possible the collimator

studied in this work is made of a tungsten alloy with a large cross section for absorption of gamma-rays. But even in this material gamma rays have a certain penetration depth, giving rise to transmission through the collimator block near the edges of the slit (where a short path length through the collimator material can be achieved). Usually this contribution to the intensity is ignored.

In this project the task is to calculate this contribution so that the tomography may be made more accurate. Staffan Jacobsson, a Ph.D. student at the group, had already made a program in Fortran90 to calculate the transmission through the collimator. In this work the program has been edited and deeper analyses have been made.

One thing to be investigated was if the gamma-ray transmission from a point in the horizontal plane is proportional to the total transmission from a vertical fuel rod. This was investigated using the quotient between the transmission from a point at $z=0$ and the total transmission.

The step length in the integration also had to be optimized. In the original configuration of the program the execution time was tree days and it was desirable to shorten it. Once the integration was optimized full calculations were done for four different energies of the gamma rays. These calculations resulted in a 3D-surface to which analytic functions were fitted with reasonable reproducibility.

2 Theory

The tomographic algorithm applied in this work is of the algebraic type. Emitted radiation is detected and the internal activity distribution is reconstructed. Such an algebraic reconstruction can be represented mathematically as a matrix of geometric coefficients times the activity vector which is equal to the intensity vector, see eq 2.1.

$$\overline{\overline{W}} \cdot \overline{A} = \overline{I} \quad \text{Eq 2.1}$$

In nuclear fuel the geometry is well known and geometric information is taken into account when calculating the matrix \overline{W} . From eq 2.1 the activity vector \overline{A} can be solved if the intensities are measured and the matrix \overline{W} is calculated from the geometric properties of the nuclear fuel. In the calculations of the coefficient matrix the collimator transmission may be included using the following integral:

$$w_{coll}(x, y) = \int_{z_{fuel}} \int_{A_{det}} e^{-\mu_{coll} d_{coll}(x, y, z, z_{det})} \partial y_{det} \partial z_{det} \partial z \quad (d_{coll} \neq 0) \quad \text{Eq 2.2}$$

μ_{coll} = the attenuation coefficient of the collimator

d_{coll} = the distance through the collimator

This integral can not be solved analytically so one must use numerical integration. To do this, the geometry of the problem has to be known.

The experimental set up consists of a collimator block of solid metal in which a slit has been made, and a detector. The detector is mounted in front of the slit going through the collimator, see figure 2.1. The radiation reaching the detector from a rod in various positions x, y was investigated. The direct radiation is the radiation passing from the fuel rod through the slit to the detector. Normally this is the only

intensity considered, i.e. one makes the approximation of an ideal collimator. In reality, gamma rays pass through short distances in a metal. This makes collimator transmission non-neglectable near the edges of the slit where the radiation can pass through the corners into the slit. The integral (Eq 2.2) represents the amount of radiation from a fuel rod in a certain position that reaches the detector by at some point passing through the collimator block.

The distance through the collimator block d_{coll} is calculated for each point in the fuel and each point on the detector. To make the integration over a fuel rod one starts in one point on the fuel rod and calculates a line from this point to a point on the detector. One then calculates how long the distance through the solid collimator block is and subtracts the distance through the slit. In this way one gets the actual distance through the collimator block. By performing calculations for a large number of points, covering the whole surface of the detector, a value of the transmission for one point in the fuel rod is obtained. Then one does the same for the points in the z -direction in the fuel and adds the contributions to a total transmission from the whole fuel rod.

In this way one gets the collimator transmission from fuel rods in various positions. This quantity is then added to the coefficient matrix used in the tomography. For symmetry reasons only points in the fuel with positive y and z coordinates have to be considered. The integration procedure is illustrated in figures 2.2-2.4.




Figure 2.1 The (simulated) geometry of the collimator.




Figure 2.2 Integration is performed over vertical rods with positive y- and z- coordinates




Figure 2.3 Integration area on the detector from one point in the fuel rod.




Figure 2.4 Successive integration in z-direction in fuel.

One of the questions to answer in this project work is whether the value in a point on the horizontal plane ($z=0$) is proportional to the total contribution from the whole fuel rod. This was evaluated using the value at $z=0$ divided by the total collimator transmission according to eq 2.2.

According to eq.2.2, the transmission depends strongly on the attenuation coefficient, i.e. on the material of the collimator and on the energy of the gamma rays.

As mentioned in section 1, the collimator is made of a tungsten alloy in order to obtain as high values of the attenuation coefficient as possible. When performing the measurements, various gamma-ray energies may be selected using spectroscopic analysis. Therefore it is of interest to calculate the collimator transmission for various energies i.e. various decays. In this work the decays of ^{140}La (1596 keV), ^{137}Cs (662 keV), ^{144}Pr (2186 keV) and ^{95}Zr (757 keV) have been studied.

3 Simulations

Some studies of the collimator transmission had been performed by Staffan Jacobsson, resulting in a program written in Fortran90 for calculating this effect. In the program eq 2.2 is calculated by numerical integration. The investigations in this work have been performed with this program and some modifications as described below.

3.1 Modification of the original program

A calculation was added to the program extracting the value of the collimator transmission from the fuel at $z=0$ and dividing it by the total transmission.

The following integration parameters were varied during the optimization of the calculation of eq 2.2:

$$\begin{aligned} \Delta z_{\text{fuel}} \\ \Delta y_{\text{detector}} \\ \Delta z_{\text{detector}} \end{aligned}$$

Δy_{fuel} was also varied even though it is not part of eq 2.2. Instead it determines the number of such integrals calculated in the y -direction.

The slit width was changed from 2 to 3 mm and the detector width was changed from 4 to 5 mm.

For the complete code see appendix A.

3.2 Reference simulation

A calculation of the collimator transmission using the energy 1596 keV (^{140}La) and the following setting where performed.

$$\begin{aligned} \text{y-step in fuel } &0,1 \text{ mm} \\ \text{z-step in fuel } &1,0 \text{ mm} \\ \text{y-step on detector } &0,01 \text{ mm} \\ \text{z-step on detector } &0,05 \text{ mm} \end{aligned}$$

This calculation took approximately three days to complete. The slit width was 2 mm and the detector width was 4mm. This calculation will here after be referred to as the “reference simulation”. This simulation included extraction of the quotient of the transmission from $z=0$ to the total transmission. The results are presented in section 4.1 and 4.2.

3.3 Optimization of numerical integration steps.

Optimization of the integration of eq 2.2 was made separately for the detector and the fuel to save calculation time.

The optimization of the integration over the detector was made in two parts. The step length in the fuel was set a priori to 0,1mm in both y- and z-direction. The integration steps on the detector were given equal lengths in y- and z-direction and varied simultaneously. This gives the optimal value of the integration step size. The results can be found in 4.3. In the second part the y direction was set to 0.1 mm and the z direction was varied. This gives a relative size between y and z direction. The large step length in y was chosen because of calculation time. Different steps in the y and z directions can be justified by

the difference in width and height of the collimator slit

The optimization of the integration step in the fuel was made similar to the optimization of the detector steps. The steps over the detector were set to 0,1 mm in both y- and z-direction to save calculation time. The step length in y-direction determines the number of points on the final curve and can be chosen arbitrary. The step length in z-direction was determined relative to the y direction as for the detector, see 4.4.

When both detector and fuel integration had been optimized, the results were combined and a final setting was found. For results see 4.5.

3.4 Optimized simulation and fitting of analytical expressions to 3D surfaces

Using the optimized settings found from section 4.5 a simulation was made with the energy 1596 keV. The simulation of the collimator transmission gave results in the form of a 3D-surface over the xy-plane. This 3D-surface consists of three main regions, slit region, penumbra region and umbra region. To these

regions analytical expressions were fitted using Microsoft's Excel.

A reconstruction of the 3D-surface using these analytical expressions was then compared to the surface of the reference.

3.6 Investigations with different gamma ray energies

Simulations were performed using the optimized settings and a correction to the dimensions of the collimator slit width and the detector width to fit the actual measurements. The slit width was changed from 2,0 mm to 3,0 mm and the

detector width was changed from 4,0 mm to 5,0 mm. The energies for ^{140}La (1596keV), ^{95}Zr (757 keV), ^{144}Pr (2186 keV) and ^{137}Cs (662 keV) were simulated

4 Results

4.1 Collimator transmission in reference simulation

The reference simulation was made using high resolving integration settings see section 3.2. This calculation took approximately tree days. The resulting values of the collimator transmission showed tree major areas: slit section, penumbra section and umbra section.

- 1 Slit section, corresponds to the area where the detector is subject to maximum direct exposure.
- 2 Penumbra section, is the part with only partial direct exposure.
- 3 Umbra section is the area where the detector is completely shielded by the collimator.

- 3 Umbra section is the area where the detector is completely shielded by the collimator.

This is illustrated in figure 4.1.1. The direction of the x-axis in figure 4.1.1 has been reversed compared to the illustrations in figure 2.1 i.e. the detector is placed to the left in the figure. (Origin is not where the system of coordinates is placed in the figure)




Figure 4.1.1

4.2 Quotient of transmission at $z_{fuel}=0$ to total

The quotient between transmission at $z=0$ and the total transmission was not constant over movements in x and y -direction, as shown in figure 4.2.1.




Figure 4.2.1

4.3 Optimization of numerical integration over detector surface

During the optimization process the energy 1596 keV was used and the distance to the detector was set to 390 mm (x value). Optimization was done over the detector in two steps. First equal steps in the y - and z -direction was applied. The step length was varied from 0,1 to 0,01 mm. This is shown for the three regions in figure 4.3.1- figure 4.3.3.




Figure 4.3.1,




Figure 4.3.2,




Figure 4.3.3

The figures above show the collimator transmission as a function of the step length for a point in each of the three sections (0,5 mm, 1,5 mm and 2,5 mm from the center of the slit). A step length of 0,05 mm or less gives approximately the same value for the collimator transmission.

In the second step the step length in y was set to 0,1 mm while the step length in z was varied. The large step in y save calculation time.




Figure 4.3.4,




Figure 4.3.5,




Figure 4.3.6

A step length of up to 0,5 mm in the z direction gives a reasonably stable value for the transmission. This gives that the step length in z can be a factor five bigger than in the y direction without losing accuracy.

4.4 Optimization of numerical integration over fuel

The step length over the detector was set to 0,1 mm in both y and z direction to save calculation time. Optimization of the step length in integration over the fuel was made in two steps. First both y and z step was set equal and varied until the curve approached the reference. The y step is not a part of eq 2.2 but it decides the resolution of the transmission profile in the y-direction.




Figure 4.4.1

The curve with step length of 0,1 mm approaches the reference. The oscillation is due to the large step lengths on the detector.

In the second part the step in y was set to 0,1 mm while the step in z was varied. The results for various steps in the fuel are represented in figure 4.4.1– 4.4.3. The figures represent points in the three different regions of the transmission.




Figure 4.4.1




Figure 4.4.2




Figure 4.4.3

A step length of 5 mm or less gives a reasonably stable value for the collimator transmission.

4.5. Selected set of numerical integration steps

The results in section 4.3 and 4.4 give the range for the integration steps over the detector and fuel separately. By combining these results a number of possible sets of integration steps were found. Compared to the reference simulation (section 4.1) the different settings gave the deviations shown in figure 4.5.1. The simulation was performed for a distance to the detector of 390 mm (x value).




Figure 4.5.1

As expected the oscillations around the reference, decrease as the step lengths are decreased. Most oscillations are gone for the setting yF0,1 zF5,0 yD0,0125 zD0,0625.

The following execution were measured during the simulations:

Settings:	Min	Sec
yF0,1 zF5,0 yD0,1 zD0,5	0	9
yF0,1 zF5,0 yD0,05 zD0,25	0	34
yF0,1 zF5,0 yD0,025 zD0,125	2	4
yF0,1 zF5,0 yD0,0125 zD0,0625	8	18

Table 4.5.1

There is no exact time for the reference (yF0,1 zF1,0 yD0,01 zD0,05) but the setting yF0,5 zF0,5 yD0,01 zD0,01 took two hours and thirty minutes to calculate. From this it is possible to estimate the calculation time for the reference to 75 minutes.

The two last settings give the best fit to the reference. It is important to observe the oscillations in the penumbra region. It is in this region that the largest deviations occur. The

oscillations occur because of the variation in step length. The smaller the step, the better the integration “sees” the edges of the slit. If the steps are too big the curve makes big jumps when the integration passes the edge.

The collimator transmission is approximately 10-20% of the total intensity. A deviation in the simulations of the collimator transmission of 2% would give a total deviation of 0,2-0,4% in the total intensity. The conclusion from this optimization is that the following settings should give reasonable accuracy while giving short calculation time.

y-step in fuel	0,1 mm
z-step in fuel	5,0 mm
y-step on detector	0,0125 mm
z-step on detector	0,0625 mm

Table 4.5.2.

4.6 Collimator transmission in optimized simulation

A complete simulation of the collimator transmission in both y and x direction was made using the settings from table 4.5.2. The step length in x direction was set to 15 mm starting at a distance of 390 mm from the detector. This step length has not been subject to investigation. The result is shown in figure 4.6.1.




Figure 4.6.1

The surface is similar to the surface in the reference simulation (section 4.1) although a slight step behavior can be noticed. The execution time for this simulation was four hours and thirty minutes compared to the execution time of tree days for the reference, implying a reduction of the execution time with 94%.

A comparison between the optimized simulation and the reference simulation gives the results shown in figure 4.6.2.




Figure 4.6.2

The difference between the two simulations were never more than 4% and usually under 2%. This is slightly more than the deviation measured in section 4.5. A deviation of 0,4-0,8 % in the total transmission is an acceptable inaccuracy with respect to the significantly shorter calculation time using optimized settings.

4.7 3D-surfaces

In these regions the curve shape in the y-direction is almost constant and changes only slightly during movements in the x-direction. This makes it possible to fit an analytical expression to the surface. The curves in the y-direction were divided into three sections representing the three regions mentioned above. Each section could be fitted with an analytical expression in the variable y. The numerical values contained in these analytical expressions were plotted as functions of x. To these curves, fits were made in the variable x. In this way an analytical expression in x and y was achieved for each section of the surface. The intersections of the different areas could be described as functions y(x). These were then used in the definitions of the limits of the different regions

A fit of an analytical function was made to the simulation in section 4.7. The selected analytical function that describes the surface is as follows:

Section	Limits	Transmission function
Slit region	$0 < y < 1.5$	$T = a_1 x + a_2$
Penumbra region	$1.5 < y < 0.0067x - 1$	$T = (b_1 x^4 + b_2 x^3 + b_3 x^2 + b_4 x + b_5)y + (c_1 x^4 + c_2 x^3 - c_3 x^2 + c_4 x + c_5)$
Umbral region	$0.0067x - 1 < y$	$T = (d_1 x^{d_2})y^{(d_3 x + d_4)}$

Table 4.7.1

$a_1 = 0,41$ $a_2 = 73,25$
 $b_1 = 2,94463e-7$ $b_2 = -6,23611422e-4$ $b_3 = 0,4955921$ $b_4 = -157,476125$ $b_5 = 23433,26$
 $c_1 = -3,34618e-7$ $c_2 = 7,0417182e-4$ $c_3 = 0,5559346$ $c_4 = 195,87527$ $c_5 = -25828,99$
 $d_1 = 1,33e-31$ $d_2 = 13,529$ $d_3 = -0,0019$ $d_4 = -7,6299$

This function was used to reconstruct the surface of the collimator transmission, resulting in the surface illustrated in figure 4.7.1




Figure 4.7.1

The similarity to the original surface is obvious. But a closer comparison with the reference (section 4.1) reveals a difference in the curvature in the umbra section as shown in figure 4.7.2.




Figure 4.8.2

The slit and penumbra regions are reproduced in good agreement with the reference. In the umbra region there is a certain deviation. No better fit was possible in this area using Microsoft's Excel to do the curve fitting, this is due to a lack of options and different curve types.

4.9 Collimator transmission for various gamma-ray energies

Simulations using four different gamma ray energies were made using the optimized integration steps accounted for in table 4.5.2. In these simulations the slit width was 3 mm and the detector width was 5 mm. The energies simulated were 1596 keV, 757keV, 2186 keV and 662 keV




Figure 4.9.1




Figure 4.9.2




Figure 4.9.3




Figure 4.9.4

As expected, the collimator transmission is higher the higher the energy is. The oscillations, appearing due to the finite integration steps, are larger for lower gamma-ray energies. This may indicate that shorter steps may be needed for lower energies.

5 Conclusion and Discussion

It can be seen from the results that the values on the horizontal plane are not representative for the total transmission. So the calculations may not be simplified by only calculating the $z=0$ value.

On the other hand the results indicate that an analytical expression describing the transmission can be applied. It is however desirable to find a better expression for the umbra region than was obtained in this work.

It can also be seen that there are some effects due to the choice of step lengths. The staircase behavior occurs because it is too far

between the points to accurately resolve the edges of the slit. So when a point turns from not “seeing” to “seeing” the edge the value at that point is drastically increased. These steps are averaged when a surface fit is made and does not influence the result very much.

Concluding the study, the optimization and calculations have been successful in the saving of time while not losing too much in accuracy. In the future it is of interest to make more calculations for other energies and to make better fits of analytical functions to the surfaces.

6 Appendix A

```
program CollimatorTransmission
!
! Created 990408 by Staffan Jacobsson.
! Latest change 010119 by Marcus Agåker. (Only transmission is calculated.)
!                                     (Quotient z0 total transmisson is calculated)
!
! Measures are in [mm].
! Origin is in collimator opening towards detector, in slit centre.
! Detector is <DetXPosition> mm behind collimator, area <DetWidth> x <DetHeight> mm.
! Collimator block is <CollLength> x <CollSide> x <CollSide> mm.
! Collimator slit is <CollLength> x <SlitWidth> x <SlitHeight> mm.
!
use Geometry_3D_mod
use Attenuation_mod
!
implicit none
!
integer,parameter::NumOfDistsToDet=15
character(len=FileNameLength),dimension(1:NumOfDistsToDet),parameter:: &
    OutFiles= (/ &
        "CollTransm_R390.out", &
        "CollTransm_R405.out", &
        "CollTransm_R420.out", &
        "CollTransm_R435.out", &
        "CollTransm_R450.out", &
        "CollTransm_R465.out", &
        "CollTransm_R480.out", &
        "CollTransm_R495.out", &
        "CollTransm_R510.out", &
        "CollTransm_R525.out", &
        "CollTransm_R540.out", &
        "CollTransm_R555.out", &
        "CollTransm_R570.out", &
        "CollTransm_R585.out", &
        "CollTransm_R600.out" /)
real(dup),parameter,dimension(1:NumOfDistsToDet):: &
    DistsFromDetector= (/ &
        390.0_dup, &
        405.0_dup, &
        420.0_dup, &
        435.0_dup, &
        450.0_dup, &
        465.0_dup, &
        480.0_dup, &
        495.0_dup, &
        510.0_dup, &
        525.0_dup, &
        540.0_dup, &
        555.0_dup, &
        570.0_dup, &
        585.0_dup, &
        600.0_dup /)
```

```

real(dup),parameter:: &
  CollSide=100.0_dup, &
  CollSideByTwo=one_half*CollSide, &
  CollLength=300.0_dup, &
  SlitHeight=30.0_dup, &
  SlitHeightByTwo=one_half*SlitHeight, &
  SlitWidth=3.0_dup, &
  SlitWidthByTwo=one_half*SlitWidth, &
  DetHeight=50.0_dup, &
  DetHeightByTwo=one_half*DetHeight, &
  DetWidth=5.0_dup, &
  DetWidthByTwo=one_half*DetWidth, &
  DetXPosition=one/one_thousand, &
  my_Coll=my_TriametS18-Cs137_662, &
  ShortestDist=0.0001_dup
integer,parameter:: &
  OutUnit=88
real(dup),parameter:: &
  yFStep=0.1 , &
  yFStart=0, &
  zFStep=5.0, &
  zFStart=one_half*zFStep, &
  yDStep=0.0125, &
  yDStart=-DetWidthByTwo+one_half*yDStep, &
  zDStep=0.0625, &
  zDStart=-DetHeightByTwo+one_half*zDStep
type(Brick_3D_type)::CollPackage,CollSlit
type(Point_3D_type)::FuelP,DetP
type(Line_3D_type)::Ray
real(dup):: &
  xF,yF,zF, &
  yFLimit,zFLimit, &
  xD,yD,zD, &
  DistInCollPack,DistInCollSlit,DistInColl, &
  Transmission,Kvotz0transmission,z0transmission
integer::LoopYF,LoopZF,LoopYD,LoopZD,LoopXF
!
!--Create collimator.
CollPackage=CREATE_BRICK_3D( &
  (-CollLength,zero,-CollSideByTwo,CollSideByTwo,-CollSideByTwo,CollSideByTwo) )
CollSlit=CREATE_BRICK_3D( &
  (-CollLength,zero,-SlitWidthByTwo,SlitWidthByTwo,-SlitHeightByTwo,SlitHeightByTwo) )
xD=DetXPosition
!
!
do LoopXF=1,NumOfDistsToDet
  xF=-DistsFromDetector(LoopXF)
  zFLimit=DetHeight*DistsFromDetector(LoopXF)/CollLength-DetHeightByTwo
  yFLimit=DetWidth*DistsFromDetector(LoopXF)/CollLength-DetWidthByTwo
  !
  open(unit=OutUnit,file=OutFiles(LoopXF),form='FORMATTED',position='APPEND')
    write(OutUnit,'(A,F8.3,A)') &
      " Distance from detector: ",DistsFromDetector(LoopXF)," mm."
    write(OutUnit,'(A,F8.3,A)') &
      " Collimator length: ",CollLength," mm."
    write(OutUnit,'(A,F8.3,A)') &

```

```

        " Collimator slit height: ",SlitHeight," mm."
write(OutUnit,'(A,F8.3,A)') &
        " Collimator slit width: ",SlitWidth," mm."
write(OutUnit,*)
write(OutUnit,'(A,F8.3,A)') &
        " Detector height: ",DetHeight," mm."
write(OutUnit,'(A,F8.3,A)') &
        " Step on detector, z-direction: ",zDStep," mm."
write(OutUnit,'(A,F8.3,A)') &
        " Detector width: ",DetWidth," mm."
write(OutUnit,'(A,F8.3,A)') &
        " Step on detector, y-direction: ",yDStep," mm."
write(OutUnit,*)
write(OutUnit,'(A)') " Integration is performed over the z-coordinate in the fuel."
write(OutUnit,'(A,F8.3,A)') &
        " From the centre (zero) to: ",zFLimit," mm."
write(OutUnit,'(A,F8.3,A)') &
        " Step in fuel, z-direction: ",zFStep," mm."
write(OutUnit,*)
write(OutUnit,'(A)') &
        " DistFromSlitCentre Transmission Kvotz0transmission"
close(OutUnit)
!
!---Loop over yFuel.
LoopYF=0
whileYF: do
    yF=yFStart+LoopYF*yFStep
    z0transmission=zero
    !---
    !---Start with transmission at z=0.
    FuelP=(/xF,yF,zero/)
    LoopYD=0
    whileYD_zF0: do
        yD=yDStart+LoopYD*yDStep
        if (yD>DetWidthByTwo) exit whileYD_zF0
        !---Loop over zDetector.
        LoopZD=0
        whileZD_zF0: do
            zD=zDStart+LoopZD*zDStep
            if (zD>DetHeightByTwo) exit whileZD_zF0
            DetP=(/xD,yD,zD/)
            Ray=(/FuelP,DetP/)
            DistInCollPack=Ray.InfLineDistWithin.CollPackage
            DistInCollSlit=Ray.InfLineDistWithin.CollSlit
            DistInColl=DistInCollPack-DistInCollSlit
            if (DistInColl>ShortestDist) then
                z0transmission=z0transmission+exp(-my_Coll*DistInColl)
            end if
            LoopZD=LoopZD+1
        end do whileZD_zF0
        LoopYD=LoopYD+1
    end do whileYD_zF0
    !---
    !---Now loop over zFuel.
    Transmission=zero
    LoopZF=0

```

```

whileZF: do
  zF=zFStart+LoopZF*zFStep
  if (zF>zFLimit) exit whileZF
  FuelP=(/xF,yF,zF/)
  !---Loop over yDetector.
  LoopYD=0
  whileYD: do
    yD=yDStart+LoopYD*yDStep
    if (yD>DetWidthByTwo) exit whileYD
    !---Loop over zDetector.
    LoopZD=0
    whileZD: do
      zD=zDStart+LoopZD*zDStep
      if (zD>DetHeightByTwo) exit whileZD
      DetP=(/xD,yD,zD/)
      Ray=(/FuelP,DetP/)
      DistInCollPack=Ray.InfLineDistWithin.CollPackage
      DistInCollSlit=Ray.InfLineDistWithin.CollSlit
      DistInColl=DistInCollPack-DistInCollSlit
      if (DistInColl>ShortestDist) then
        Transmission=Transmission+exp(-
my_Coll*DistInColl)
      end if
      LoopZD=LoopZD+1
    end do whileZD
    LoopYD=LoopYD+1
  end do whileYD
  LoopZF=LoopZF+1
end do whileZF
!
!---Scale according to steps in various directions.
Transmission=Transmission*yDStep*zDStep*zFStep
z0transmission=z0transmission*yDStep*zDStep
Kvotz0transmission=z0transmission/Transmission
!
open(unit=OutUnit,file=OutFiles(LoopXF),form='FORMATTED',position='APPEND')
write(OutUnit,'(F8.3,2F12.5)') &
  yF,Transmission,Kvotz0transmission
close(OutUnit)
!
if (yF>yFLimit) exit whileYF
LoopYF=LoopYF+1
!
end do whileYF
end do
!
end program CollimatorTransmission

```